

## STAT 5200 Handout #18

### Randomized Complete Block Design [RCBD] (Ch. 13)

#### Motivating Example: Turkey Diets

An experiment is carried out to evaluate the effects of two dietary factors on the growth of turkeys. The two dietary factors are Sodium and Chloride. Both of these factors have two levels—"High" and "Low"—so there are four combinations of factor levels. Four buildings are available for the experiment. Each building contains four pens. Within each building, the four combinations of Sodium and Chloride are randomly assigned to the pens. At the end of the experiment 30 turkeys are randomly selected from each pen and the average weight of the 30 turkeys is recorded for the corresponding pen.

Building:	1				2				3				4			
Sodium:	H	H	L	L	H	H	L	L	H	H	L	L	H	H	L	L
Chloride:	H	L	H	L	H	L	H	L	H	L	H	L	H	L	H	L
Pen:	1	4	3	2	6	8	5	7	12	9	10	11	13	16	15	14

CRD within
CRD within
---
} 4 separate randomizations

What if we fit this as a CRD (completely randomized design), with experimental units (Pens) having been randomly assigned to treatments (Sodium-Chloride [factorial] combinations)




```

data TurkeyDiets;
  input Building Pen BodyWeight Sodium $ Chloride $ @@;
  cards;
  1 1 969 High High 2 6 928 High High
  3 12 903 High High 4 13 961 High High
  1 4 1036 High Low 2 8 885 High Low
  3 9 959 High Low 4 16 1067 High Low
  1 3 1046 Low High 2 5 970 Low High
  3 10 931 Low High 4 15 1038 Low High
  1 2 1021 Low Low 2 7 883 Low Low
  3 11 924 Low Low 4 14 932 Low Low
  ;


proc glm data=TurkeyDiets plots=diagnostics;
  class Sodium Chloride;
  model BodyWeight = Sodium|Chloride;
  lsmeans Sodium*Chloride;
  title1 'Completely Randomized Design';
run;
    
```

Completely Randomized Design

Source	DF	Type III SS	Mean Square	F Value	Pr > F
Sodium	1	85.56250	85.56250	0.02	0.8785
Chloride	1	95.06250	95.06250	0.03	0.8720
Sodium*Chloride	1	10557.56250	10557.56250	3.01	0.1083



Interaction Plot for BodyWeight



What are we missing?

→ there is a restriction on randomization (unaccounted for)

This is actually not a simple completely randomized design.

Instead, it is a **randomized complete block design (RCBD)** because experimental units (Pens) are **randomly assigned** to treatments (Sodium-Chloride [factorial] combinations) **within blocks** (Building).

*a natural or pre-made grouping of exp. units*  
**General Discussion**

Sometimes experimental units are already in natural (& homogeneous) blocks (or groupings) such as:

- Batches of raw material (from supplier; sometimes litter [rats])
- Spatial groupings (fields or neighborhoods)
- Temporal groupings (week or day)
- Pens in Buildings

We can use these blocks (and their natural homogeneity) :

*when we account for blocks*

1. Variance reduction (another term for “blocking”, or using blocks in a design)
  - We get a better estimate of  $\sigma^2$  (error variance, or variance among “homogeneous” experimental units; “homogeneous” because after accounting for trt. differences, there is no other real difference among units)  
 *$\hat{\sigma}^2 = MSE$*
2. Increase statistical power
  - Probability of detecting real treatment differences

Note: Blocking (or block structure) is an inherent characteristic of the experimental units

*- not something we impose on them*

Key features of a randomized complete block design (RCBD):

- Randomly assign experimental units to treatments *within blocks*
  - so all treatments are used in all blocks or comb. of factor levels
  - a completely randomized design (CRD) is run within each block
- Each randomization is independent of others  
*(within block)*

**Back to Example: Turkey Diets**

Model:

$$Y_{ijk} = \mu + S_i + C_j + SC_{ij} + B_k + \epsilon_{ijk}$$

$i=1,2$  Sodium  
 $j=1,2$  Chloride  
 $k=1..4$  Building

Fixed effects, usual sum-to-zero constraints  
 Random blocking term  
 $\epsilon_{ijk} \text{ iid } N(0, \sigma^2)$

Ave. weight of 30 turkeys from pen

Although we use Block in the model, we will ignore inference on Block

- We did not [randomly] assign experimental units to blocks

they (exp. units & blocks) were already there

- We just blocked to make experimental units different between blocks

- Block is just an [extraneous] source of variation


→ not really interesting, but we still need to account for it

Why no interaction with Block?

- Primarily because we ignore Block-related inference

- Also, that interaction is our error term!

Hasse Diagram:






MODEL BodyWeight = Sodium|Chloride;  
 RANDOM Building;

actually a random term, but this choice (fixed vs. random) for Building won't affect any tests here, where all F-tests will be

```
proc mixed data=TurkeyDiets plot=residualpanel method=type3;
  class Building Sodium Chloride;
  model BodyWeight = Building Sodium|Chloride;
  title1 'Randomized Complete Block Design';
run;
/* NOTE: could also do post-hoc mean tests, contrasts,
and interaction plots where appropriate */
```

$F = \frac{MS_{\text{term}}}{MS_E}$   
 (E is next random term below everything whether B is fixed or random)  
 because of a strong block effect  
 now more signif.

Type 3 Tests of Fixed Effects				
Effect	Num DF	Den DF	F Value	Pr > F
Building	3	9	7.91	0.0068
Sodium	1	9	0.07	0.8022
Chloride	1	9	0.07	0.7918
Sodium*Chloride	1	9	8.21	0.0186



Residual Statistics	
Observations	16
Minimum	-52.44
Mean	0
Maximum	46.563
Std Dev	27.774
Fit Statistics	
Objective	98.29
AIC	100.29
AICC	100.86
BIC	100.49